

	******************************************************************
	**
	**
	**		NAME:	ELITE PERSISTENCE IN THE ERA OF EXPANDING OVERSEAS TRADE
	**
	**		MAKES:  Figures 3 (a) and (b)		       
	**		
	**
	******************************************************************
	
	
	
	
	

			

		
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* global for output
*-------------------------------------------------------------------------------


* input
*------

global input = "" 	
	
	
* output
*-------

global output = ""	
		
	
	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* preliminaries
*-------------------------------------------------------------------------------


* preliminaries
*--------------

clear
set more off




			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* data
*-------------------------------------------------------------------------------


* use
*----

cd ${input}
use analysis_vars_parl.dta, clear 




			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* assert
*-------------------------------------------------------------------------------


* assert
*-------

assert 	parl_year >= 1536 & parl_year <= 1760


duplicates tag parl_group, gen(tt)
assert tt == 0 if parl_group != .
drop tt






			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* globals
*-------------------------------------------------------------------------------


* globals
*--------

global yr1 		= "parl_year > 1553 & parl_year < 1639"
global yr2 		= "parl_year >= 1660 & parl_year <= 1756"

global yr3 		= "parl_year >= 1639 & parl_year <= 1649"
global yr4 		= "parl_year >= 1628 & parl_year <= 1660"

global yr5 		= "parl_year < 1639 & parl_year >= 1583"
global yr6 		= "parl_year > 1550 & parl_year <= 1750"

global yr7 		= "parl_year > 1553 & parl_year <= 1640"


global xline1	= "xline(1660, lcolor(gs12))"
global xline2	= "xline(1688, lcolor(gs12) lpattern(line) )"
global xline3	= "xline(1640, lcolor(gs12))"
global xline4	= "xline(1562, lcolor(gs12)  lpattern(line) )"


	
	




			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**


*-------------------------------------------------------------------------------
* PLOT:  active merchants by geography (FIGURE 3 (a))
*-------------------------------------------------------------------------------



* hi lo
*------

capture drop hi 
capture drop lo
gen hi = 9
gen lo = 0
sort parl_group



* this plot
*----------

global outcome1 = "int_intl_east_pperc"
global outcome2 = "int_intl_newworld_pperc"
  
gen ${outcome1}_pct = ${outcome1} * 100
gen ${outcome2}_pct = ${outcome2} * 100

global scale = "0(2)9"

  

* plot
*-----

# delimit ;

twoway 
	
	
		// shaded area
	(rarea lo hi parl_year	
		if   parl_year >= 1639 & parl_year <= 1649 
		,  color(gs12) )
		
			
		
	(lpoly ${outcome1}_pct parl_year
		if   ${yr1}		
		,  bwidth(8) lcolor(gs1)  lwidth(medthick) lpattern(line) ${xline2} ${xline4} )
	(lpoly ${outcome1}_pct parl_year
		if   ${yr2}		
		,  bwidth(8) lcolor(gs1)  lwidth(medthick) lpattern(line)  )	
			
	(lpoly ${outcome2}_pct parl_year
		if   ${yr1}		
		,  bwidth(8) lcolor(gs5)  lwidth(medthick) lpattern(shortdash) ${xline2} ${xline4} )
	(lpoly ${outcome2}_pct parl_year
		if   ${yr2}		
		,  bwidth(8) lcolor(gs5)  lwidth(medthick) lpattern(shortdash)  )
		
		
		
	(scatter ${outcome1}_pct parl_year
		if    ${yr1}		
		,  mcolor(gs1) msize(medlarge) msymbol(circle))
	(scatter ${outcome1}_pct parl_year
		if    ${yr2}			
		,  mcolor(gs1) msize(medlarge) msymbol(circle))	
		
		
	(scatter ${outcome2}_pct parl_year
		if    ${yr1}		
		,  mcolor(gs5) msize(medlarge) msymbol(circle_hollow))
	(scatter ${outcome2}_pct parl_year
		if    ${yr2}			
		,  mcolor(gs5) msize(medlarge) msymbol(circle_hollow))		
		
		,
		yscale(noline)
		xscale(noline)
		

		
		xsize(8.5)
		ysize(3)

		title("", color(black) size(medsmall) pos(11) ) 
		xlab(  1550(25)1750, 
			labsize(medsmall) nogrid ) 
		ylab( ${scale}, 
			angle(hori) labsize(medsmall) nogrid  )
		xtitle("Year" , 
			color(black) size(large) )
		ytitle("Percentage of MPs", 
			color(black) size(large)  )
		graphregion( fcolor(white) lcolor(white) ) 
		plotregion( fcolor(white) lstyle(none) lcolor(white) ilstyle(none)) 
		legend(order(				
				2 "% MPs w/ Overseas Atlantic Trade Occupation to Asia"
				4 "% MPs w/ Overseas Atlantic Trade Occupation to the Americas"

				)
			cols(1)
			pos(6)
			
			region( color(none) )
			size(large)
			)
		
		;

# delimit cr


	* export graph
	*-------------

	cd "${output}"
	graph export "trend_commercial_active.pdf", replace
	

	
	
	
	
	
	

			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**


*-------------------------------------------------------------------------------
* PLOT:  passive investors by geography of trade (FIGURE 3 (b))
*-------------------------------------------------------------------------------



* hi lo
*------

capture drop hi 
capture drop lo
gen hi = 21
gen lo = 0
sort parl_group



* this plot
*----------


global outcome1 = "pass_asia_i_pperc"
global outcome2 = "pass_newworld_i_pperc"

gen ${outcome1}_pct = ${outcome1} * 100
gen ${outcome2}_pct = ${outcome2} * 100
  
global scale = "0(5)20"

  

* plot
*-----

# delimit ;

twoway 
	
	
		// shaded area
	(rarea lo hi parl_year	
		if   parl_year >= 1639 & parl_year <= 1649 
		,  color(gs12) )
		
		

		
	(lpoly ${outcome1}_pct parl_year
		if   ${yr1}		
		,  bwidth(8) lcolor(gs1)  lwidth(medthick) lpattern(line) ${xline2} ${xline4} )
	(lpoly ${outcome1}_pct parl_year
		if   ${yr2}		
		,  bwidth(8) lcolor(gs1)  lwidth(medthick) lpattern(line)  )	
		
	(lpoly ${outcome2}_pct parl_year
		if   ${yr1}		
		,  bwidth(8) lcolor(gs5)  lwidth(medthick) lpattern(shortdash) ${xline2} ${xline4} )
	(lpoly ${outcome2}_pct parl_year
		if   ${yr2}		
		,  bwidth(8) lcolor(gs5)  lwidth(medthick) lpattern(shortdash)  )	
		


	(scatter ${outcome1}_pct parl_year
		if    ${yr1}		
		,  mcolor(gs1) msize(medlarge) msymbol(circle))
	(scatter ${outcome1}_pct parl_year
		if    ${yr2}			
		,  mcolor(gs1) msize(medlarge) msymbol(circle))		
	

	(scatter ${outcome2}_pct parl_year
		if    ${yr1}		
		,  mcolor(gs5) msize(medlarge) msymbol(circle_hollow))
	(scatter ${outcome2}_pct parl_year
		if    ${yr2}			
		,  mcolor(gs5) msize(medlarge) msymbol(circle_hollow))	
		
		
		,
		yscale(noline)
		xscale(noline)
		
		
		xsize(8.5)
		ysize(3)

		title("", color(black) size(medsmall) pos(11) ) 
		xlab( 1550(25)1750, 
			labsize(medsmall) nogrid) 
		ylab( ${scale}, 
			angle(hori) labsize(medsmall) nogrid  )
		xtitle("Year" , 
			color(black) size(large) )
		ytitle("Percentage of MPs", 
			color(black) size(large)  )
		graphregion( fcolor(white) lcolor(white) ) 
		plotregion( fcolor(white) lstyle(lgrey) lcolor(white) ilstyle(none)) 
		legend(order(				
				2 "% MPs w/ Overseas Atlantic Trade to Asia Investment"
				4 "% MPs w/ Overseas Atlantic Trade to the Americas Investment"
				)
			cols(1)
			pos(6)
			
			region( color(none) )
			size(large)
			)
		
		;

# delimit cr


	* export graph
	*-------------

	cd "${output}"
	graph export "trend_commercial_passive.pdf", replace
	
	
	
	
	
	
		
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	

						** end of do file ** 	
	
	
	